

The background is a dark blue gradient with a subtle pattern of white dots. Overlaid on the left side are several white circular and semi-circular lines, some with arrows indicating a clockwise direction. A large, semi-circular scale is visible, with numerical markings ranging from 140 to 260 in increments of 10.

STATA 数据分析结果导出

张逸凡 z.yf@pku.edu.cn

最后更新：2021-02-14

0. 总览

概览本文档的结构和主要内容。

如何将数据分析结果
导出至文件

tabout 命令

高质量描述统计表格

estout 命令

回归分析表格

outreg2 命令

高质量回归分析表格

asdoc 命令

doc格式的各种表格

sum2docx, corr2docx, reg2docx 命令

docx格式描述统计/相关性分析/回归分析表格

graph export 命令

图片

1. tabout 命令

导出多种格式的精致的描述统计表格。

1. tabout 命令

- 什么是tabout? 官方介绍:
 - “tabout是一个用于制作印刷质量的描述统计表格的程序”;
 - “Ben Jann编写的estout程序.....专门制作基于估计的表格, 如回归结果表, 而tabout则专注描述统计”。
- 官网: <http://tabout.net.au/docs/home.php>
- 可从官网下载PDF格式的英文版使用指南备查。
- tabout生成的表格对中文的支持非常出色。
- tabout第二版安装: `ssc install tabout, replace`
- 如果要实现输出Word文档和输出Excel表格的功能, 必须下载安装第三版。

1. about 命令：第三版的安装


- 与第二版不同，about第三版需要手动安装。
 - 1. 在官网选择合适的版本（如图）。
 - 2. 点进去以后是一个txt文档，把它保存到本地。
 - 3. 将保存的“.txt”文档的后缀名手动替换成“.ado”。
 - 4. STATA命令行运行“adopath”，查看ado路径（如图）。
 - 5. 把重命名过的ado文件移动到指定路径（如图）。
 - 若之前未安装过第二版，则移动到personal路径；
 - 若之前已安装过第二版，则移动到plus路径。
 - 6. STATA命令行运行“which about”查看当前版本（如图）。
- 更多细节参见使用指南的“Installation”一节。

- Download **about** ado file for version 3 beta. Pick the **about** download which matches your version of Stata:

- [For Stata 14.2 or later](#)
- [For Stata 13.1](#)
- [For versions of Stata earlier than 13](#)

```
. adopath
[1] (BASE)      "C:\Program Files (x86)\Stata15\ado\base/"
[2] (SITE)      "C:\Program Files (x86)\Stata15\ado\site/"
[3]             "."
[4] (PERSONAL)  "c:\ado\personal/"
[5] (PLUS)      "c:\ado\plus/"
[6] (OLDPLACE)  "c:\ado/"
```

此电脑 > Windows (C:) > ado > plus

名称	修改日期	类型
 tabout.ado	2020/12/5 22:48	ADO 文件

```
. which about
c:\ado\plus\tabout.ado
*! Version 3.0.9 beta Ian Watson 17apr2019
*! Stata 14.2 (or later) version
```

1. tabout 命令：输出格式

- tabout 支持的输出格式包括：
 - 纯文本 (.txt)
 - LaTeX (.tex)
 - Word (.docx)
 - Excel (.xlsx)
 - 网页 (.htm)
- 以下仅演示输出Excel格式的代码。
- 输出其他格式的代码参见使用指南。

1. tabout 命令：一维描述统计表（频数）

`sysuse bpwide, clear` // 这是一个系统自带的模拟血压数据库

`global PATH = "xxx"` // 这里输入自己的工作路径

`tabout agegrp using "$PATH/表1_年龄段分布.xlsx", ///`

`replace style(xlsx) font(bold) oneway c(freq col cum) ///`

`f(0c 1) clab(数量 比例_% 累计比例_%) npos(col) ///`

`title(表1: 年龄段分布)`

1. tabout 命令：一维描述统计表（频数）

	A	B	C	D	E
1	表1：年龄段分布				
2					
3		数量	比例 %	累计比例 %	N
4	30-45	40	33.3	33.3	40
5	46-59	40	33.3	66.7	40
6	60+	40	33.3	100.0	40
7	Total	120	100.0		120
8					

1. tabout 命令：一维描述统计表（频数）

- **tabout**: 后面紧跟要做描述统计的变量名。
 - **tabout**后也可以跟多个变量，其效果是对多个变量分别进行描述统计，并聚合在一张表里。
- **using**: 后面紧跟导出的表格的文件路径。
- **replace**: 覆盖同名文件。
- **style(xlsx)**: 指定文件的输出格式，如**xlsx**、**tex**等。
- **font(bold)**: 将表中的标题、变量标签等要素粗体显示。
- **oneway**: 表明要生成一维描述统计表。
 - 官方解释是“所有变量都是‘纵轴’变量，因此不生成交叉表”。
- **c(freq col cum)**: **c**指**contents**，指定表格内容。
 - 这里列出了频数、列比例和累计比例。

1. tabout 命令：一维描述统计表（频数）

- `f(0c 1)`：f指format，每个参数控制一列。
 - c代表逗号分位符，数字代表小数位数。
 - 参数的数量比列数少时，最后一个参数控制剩下的列。
 - 本例中，即指第1列保留整数、使用千分位符，第2、3列保留1位小数、不使用千分位符。
- `clab(数量 比例_% 累计比例_%)`：clab指column label。每个参数控制一列的表头。
 - 下划线代表空格。
 - 参数的数量比列数少时，最后一个参数控制剩下的列。
- `npos(col)`：指定样本量的显示位置。col表示单辟一列显示样本量。
- `title(表1：年龄段分布)`：指定表的标题。
- 更多选项见英文版指南。

1. tabout 命令：一维描述统计表（频数）

```
sysuse bpwide, clear // 这是一个系统自带的追踪调查数据库  
global PATH = "xxx" // 这里输入自己的工作路径
```

```
tabout agegrp sex using "$PATH/表2_性别与年龄段分布.xlsx", ///  
replace style(xlsx) font(bold) oneway c(freq col cum) ///  
f(0c 1) clab(数量 比例_% 累计比例_%) npos(col) ///  
title(表2: 性别与年龄段分布)
```

1. tabout 命令：一维描述统计表（频数）

	A	B	C	D	E
1	表2：年龄段与性别分布				
2					
3		数量	比例 %	累计比例 %	N
4	Age Group				
5	30-45	40	33.3	33.3	40
6	46-59	40	33.3	66.7	40
7	60+	40	33.3	100.0	40
8	Total	120	100.0		120
9	Sex				
10	Male	60	50.0	50.0	60
11	Female	60	50.0	100.0	60
12	Total	120	100.0		120
13					

1. tabout 命令：一维描述统计表（摘要）

```
sysuse nlsw88, clear
```

```
global PATH = "xxx" // 这里输入自己的工作路径
```

```
tabout occupation race using "$PATH/表3_职业、种族与薪资、工时交叉表.xlsx", ///
```

```
replace style(xlsx) font(bold) ///
```

```
c(mean wage mean hours median wage median hours) ///
```

```
h2(均值 中位数) h2c(2 2) f(2 1 2 1) sum ///
```

```
clab(薪资（美元/小时） 工时（小时/周） 薪资（美元/小时） 工时（小时/周）) ///
```

```
lwidth(20) cwidth(17) units(cm) ///
```

```
title(表3：职业、种族与薪资、工时交叉表)
```


1. tabout 命令：一维描述统计表（摘要）

	A	B	C	D	E
1	表3：职业、种族与薪资、工时交叉表				
2					
3		均值		中位数	
4		薪资 (美元/小时)	工时 (小时/周)	薪资 (美元/小时)	工时 (小时/周)
5	occupation				
6	Professional/technical	10.72	37.8	9.68	40.0
7	Managers/admin	10.90	43.0	9.49	40.0
8	Sales	7.15	35.9	6.05	40.0
9	Clerical/unskilled	8.52	35.0	5.13	40.0
10	Craftsmen	7.15	40.4	5.80	40.0
11	Operatives	5.65	39.4	5.02	40.0
12	Transport	3.20	25.6	2.94	24.0
13	Laborers	4.91	32.0	4.06	35.0
14	Farmers	8.05	40.0	8.05	40.0
15	Farm laborers	3.08	40.6	3.02	40.0
16	Service	5.99	36.4	5.92	40.0
17	Household workers	6.39	36.0	6.39	36.0
18	Other	8.84	40.6	8.37	40.0
19	Total	7.78	37.2	6.28	40.0
20	race				
21	white	8.08	36.9	6.55	40.0
22	black	6.84	38.1	5.43	40.0
23	other	8.55	36.8	7.56	40.0
24	Total	7.77	37.2	6.27	40.0
25					

1. tabout 命令：一维描述统计表（摘要）

- `tabout`后紧跟的是“纵轴”变量，可以有多个。
- `c(mean wage mean hours median wage median hours)`：设置要显示摘要的变量。
 - 两个参数一组，使用“统计量 + 空格 + 变量名”的格式。
 - 如`mean wage`即表示薪资的均值；`median hours`即表示工时的中位数。
- `h2(均值 中位数)`：设置二级表头（在本例中改成`h1`也可以）。必须与`h2c()`命令搭配。
- `h2c(2 2)`：设置`h2`的每个表头分别管理几列。
 - 在本例中，即“均值”和“中位数”分别对应2列。
- `sum`：指`summary table`，即要生成包含均值/中位数/最值等统计量的摘要表，与频数表区分。

1. tabout 命令：二维交叉表（频数）

```
sysuse nlsw88, clear
```

```
global PATH = "xxx" // 这里输入自己的工作路径
```

```
tabout occupation race married ///
```

```
using "$PATH/表4_职业、种族与婚姻情况交叉表.xlsx", ///
```

```
replace c(freq col cum) f(0c 1) clab(数量 比例_% 累计比例_%) ///
```

```
lwidth(20) cwidth(10) units(cm) ///
```

```
font(bold) style(xlsx) title(表4：职业、种族与婚姻情况交叉表)
```


1. tabout 命令：二维交叉表（频数）

[illegible]

1. tabout 命令：二维交叉表（频数）

- `tabout`后面紧跟变量名，前一个是“纵轴”，后一个是“横轴”。
 - 如果输入多个变量，则只有最后一个是“横轴”变量，前面的都是“纵轴”变量。
- `lwidth(15) cwidth(8) units(cm)`: Excel表格中设置列宽的选项。
 - `lwidth`指label width，用于设置首列宽度；
 - `cwidth`指column width，用于设置其余列的宽度；
 - `units`设置长度参数的单位。
- `font(italic)`: 将表中的标题、变量标签等要素斜体显示。

1. tabout 命令：二维交叉表（摘要）

```
sysuse nlsw88, clear
```

```
global PATH = "xxx" // 这里输入自己的工作路径
```

```
tabout occupation race ///
```

```
using "$PATH/表5_职业与种族交叉分类下的薪资情况表.xlsx", ///
```

```
replace h1(不同种族的平均薪资（美元/小时）) h1c(4) h3(nil) ///
```

```
c(mean wage) f(2) sum ///
```

```
font(bold) style(xlsx) title(表5：职业与种族交叉分类下的薪资情况表)
```


1. tabout 命令：二维交叉表（摘要）

	A	B	C	D	E
1	表5：职业与种族交叉分类下的薪资情况表				
2					
3		不同种族的平均薪资（美元/小时）			
4	occupation	white	black	other	Total
5	Professional/technical	10.58	11.11	12.35	10.72
6	Managers/admin	11.06	9.71	10.97	10.90
7	Sales	7.13	7.14	8.84	7.15
8	Clerical/unskilled	8.47	8.90		8.52
9	Craftsmen	6.96	7.49		7.15
10	Operatives	6.04	5.33	2.47	5.65
11	Transport	3.71	2.82		3.20
12	Laborers	5.21	4.40	4.69	4.91
13	Farmers	8.05			8.05
14	Farm laborers	3.74	2.76		3.08
15	Service	5.76	6.38		5.99
16	Household workers	6.39			6.39
17	Other	8.72	9.25		8.84
18	Total	8.10	6.85	8.55	7.78
19					

1. tabout 命令：二维交叉表（摘要）

- 与一维摘要表不同，二维摘要表的`tabout`后面接的依次是“纵轴”变量和“横轴”变量。
 - 如果输入多个变量，则只有最后一个是“横轴”变量，前面的都是“纵轴”变量。
- `h3(nil)`：`nil`参数表示隐藏三级表头。
 - `c()`的参数只有一个，即平均薪资，所以直接在一级表头写明“不同种族的平均薪资”。
 - 此时`h3()`的默认值为“mean wage”，和`h1`重复，稍显累赘，所以可以隐去。

1. tabout 命令：回顾四种描述统计表

- tabout 将描述统计表按照“一维/二维”和“频数/摘要”两个标准分为四大类。
 - 一维频数表：只有“纵轴”变量，单元格里是该变量不同取值的频数、比例等统计量。
 - 一维摘要表：只有“纵轴”变量，单元格里是在该变量取不同值时，其他变量的均值、中位数等统计量。
 - 二维频数表：有“纵轴”和“横轴”变量，单元格里是纵轴和横轴分别取不同值时，交叉分类的频数、比例等统计量。
 - 二维摘要表：有“纵轴”和“横轴”变量，单元格里是纵轴和横轴交叉分类的前提下，其他变量的均值、中位数等统计量。
- “纵轴”变量最好是定类/定序变量。

1. tabout 命令：相关性分析表

```
sysuse nlsw88, clear
global PATH = "xxx" // 这里输入自己的工作路径
svyset _n
label define c_city_label 0 "No" 1 "Yes"
label value c_city c_city_label
```

- 使用**tabout**做相关性分析，首先要将数据集声明为调查集**survey set**。
- **svyset _n**：声明调查集的PSU（初级抽样单位），此处**_n**指PSU为单个样本，即简单随机抽样。

1. tabout 命令：相关性分析表

```
tabout race collgrad c_city ///  
using "$PATH/表6_居住地与种族、学历的相关性分析表.xlsx", ///  
replace style(xlsx) font(bold) ///  
c(row ci) svy stats(chi2) f(3) csep(-) level(99) ///  
clab(频数 99%置信区间) cwidth(15) indent(0) ///  
npos(col) nlab(样本量) h2c(2 2 1) ///  
title(表6: 居住地与种族、学历的相关性分析表)
```

1. tabout 命令：相关性分析表

	A	B	C	D	E	F	G	H
1	表6：居住地与种族、学历的相关性分析表							
2								
3		lives in central city						
4		No		Yes		Total	样本量	
5		频率	99%置信区间	频率	99%置信区间	频率		
6	race							
7	white	0.779	[0.751-0.804]	0.221	[0.196-0.249]	1.000	1,637	
8	black	0.509	[0.456-0.563]	0.491	[0.437-0.544]	1.000	583	
9	other	0.731	[0.465-0.895]	0.269	[0.105-0.535]	1.000	26	
10	Total	0.708	[0.683-0.732]	0.292	[0.268-0.317]	1.000	2,246	
11	Pearson: Uncorrected chi2(2)= 151.129							
12	Design-based F(2.00, 4490.00)= 75.531							
13	P-value= 0.000							
14	college graduate							
15	not college grad	0.725	[0.697-0.752]	0.275	[0.248-0.303]	1.000	1,714	
16	college grad	0.654	[0.599-0.705]	0.346	[0.295-0.401]	1.000	532	
17	Total	0.708	[0.683-0.732]	0.292	[0.268-0.317]	1.000	2,246	
18	Pearson: Uncorrected chi2(1)= 9.926							
19	Design-based F(1.00, 2245.00)= 9.922							
20	P-value= 0.002							
21								
22								

1. tabout 命令：相关性分析表

- `c(row ci)`: row指row percentage, 即列比例; ci指confidence intervals, 即置信区间。
- `svy`: 指survey, 表示要分析的是一份调查集。
- `stats(chi2)`: 在表末附上卡方检验结果。
- `cisep(-)`: 指how the confidence intervals are separated, 用于设定置信区间上界和下界之间的分隔符, 参数可以设置为任意字符。
- `level(99)`: 用于指定置信水平, 默认值为95, 即95%置信水平。
- tabout的更多功能和选项见英文版使用指南。

2. estout 命令

导出回归分析表格。

2. estout 命令

- 什么是estout？官方介绍：
 - “estout包提供了在Stata中制作回归表格的工具”。
 - “estout从之前拟合并存储的一个或多个模型中组合出一个回归表”。
- 官网：<http://repec.sowi.unibe.ch/stata/estout/>
- 五个主要命令：esttab、estout、eststo、estadd、estpost。
- 安装：`ssc install estout, replace`
- 调用帮助文档：`help estout`
 - 将estout换成esttab、eststo、estadd或estpost还可获得其他四个命令的帮助。

2. estout 命令之 esttab: 打印基本的回归分析表

```
sysuse auto, clear
```

```
eststo: quietly regress price weight mpg
```

```
eststo: quietly regress price weight mpg length
```

```
eststo: quietly regress price weight mpg length foreign
```

```
esttab
```

- 基本回归分析表，缺少一些必要的描述性数据。
- 使用 `eststo clear` 清除缓存的回归结果，否则输出的表越来越长。

	(1) price	(2) price	(3) price
weight	1.747** (2.72)	4.365*** (3.74)	5.716*** (5.63)
mpg	-49.51 (-0.57)	-86.79 (-1.03)	-13.41 (-0.19)
length		-104.9* (-2.64)	-92.48** (-2.75)
foreign			3550.2*** (5.42)
_cons	1946.1 (0.54)	14542.4* (2.47)	5515.6 (1.05)
N	74	74	74

t statistics in parentheses
* p<0.05, ** p<0.01, *** p<0.001

2. estout 命令之 esttab: 加入回归的其他返回值

```
sysuse auto, clear
```

```
eststo: quietly regress price weight mpg
```

```
eststo: quietly regress price weight mpg length
```

```
eststo: quietly regress price weight mpg length foreign
```

```
esttab, stats(r2 N, labels("R Square" "Num of Obs"))
```

- 加入R方和样本量。还可改成r2_a等。参见: `help estout##stats`
- 可以在“labels”选项里自己设置返回值的变量标签 (如图)。

	(1) price	(2) price	(3) price
weight	1.747** (2.72)	4.365*** (3.74)	5.716*** (5.63)
mpg	-49.51 (-0.57)	-86.79 (-1.03)	-13.41 (-0.19)
length		-104.9* (-2.64)	-92.48** (-2.75)
foreign			3550.2*** (5.42)
_cons	1946.1 (0.54)	14542.4* (2.47)	5515.6 (1.05)
R Square	0.293	0.357	0.549
Num of Obs	74	74	74
t statistics in parentheses			
* p<0.05, ** p<0.01, *** p<0.001			

R方	0.293	0.357	0.549
样本量	74	74	74

2. estout 命令之 esttab: 加入回归的其他返回值

- `stats()` 选项其实是 `estout` 命令提供的, 但它也能为 `esttab` 提供定制服务。
- `esttab` 自身的输出方式: `esttab`, `ar2` (左图)
- 借用 `stats()` 的输出方式: `esttab`, `stats(r2_a N)` (右图)

	(1) price	(2) price	(3) price
weight	1.747** (2.72)	4.365*** (3.74)	5.716*** (5.63)
mpg	-49.51 (-0.57)	-86.79 (-1.03)	-13.41 (-0.19)
length		-104.9* (-2.64)	-92.48** (-2.75)
foreign			3550.2*** (5.42)
_cons	1946.1 (0.54)	14542.4* (2.47)	5515.6 (1.05)
N	74	74	74
adj. R-sq	0.273	0.330	0.523

t statistics in parentheses
* p<0.05, ** p<0.01, *** p<0.001

	(1) price	(2) price	(3) price
weight	1.747** (2.72)	4.365*** (3.74)	5.716*** (5.63)
mpg	-49.51 (-0.57)	-86.79 (-1.03)	-13.41 (-0.19)
length		-104.9* (-2.64)	-92.48** (-2.75)
foreign			3550.2*** (5.42)
_cons	1946.1 (0.54)	14542.4* (2.47)	5515.6 (1.05)
r2_a	0.273	0.330	0.523
N	74	74	74

t statistics in parentheses
* p<0.05, ** p<0.01, *** p<0.001

2. estout 命令之 esttab: 显示变量标签

- 方式一: `esttab`, `ar2 label` (左图)
- 方式二: `esttab`, `stats(r2_a N, labels("Adj. R-squ." "Num of Obs")) label` (右图)
- 可见`stats()`选项提供的标签会覆盖`esttab`自带的`label`。

	(1) Price	(2) Price	(3) Price
Weight (lbs.)	1.747** (2.72)	4.365*** (3.74)	5.716*** (5.63)
Mileage (mpg)	-49.51 (-0.57)	-86.79 (-1.03)	-13.41 (-0.19)
Length (in.)		-104.9* (-2.64)	-92.48** (-2.75)
Car type			3550.2*** (5.42)
Constant	1946.1 (0.54)	14542.4* (2.47)	5515.6 (1.05)
Observations	74	74	74
Adjusted R-squared	0.273	0.330	0.523

t statistics in parentheses
* p<0.05, ** p<0.01, *** p<0.001

	(1) Price	(2) Price	(3) Price
Weight (lbs.)	1.747** (2.72)	4.365*** (3.74)	5.716*** (5.63)
Mileage (mpg)	-49.51 (-0.57)	-86.79 (-1.03)	-13.41 (-0.19)
Length (in.)		-104.9* (-2.64)	-92.48** (-2.75)
Car type			3550.2*** (5.42)
Constant	1946.1 (0.54)	14542.4* (2.47)	5515.6 (1.05)
Adj. R-squ.	0.273	0.330	0.523
Num of Obs	74	74	74

t statistics in parentheses
* p<0.05, ** p<0.01, *** p<0.001

2. estout 命令之 esttab: 输出至文件

```
global PATH = "xxx" // 输入自己的工作路径
```

```
esttab using "$PATH/estout_tab_1.xls", ar2 label
```

- 这个命令输出的表格还是无法使用。
- 如何输出更好的表格?

	A	B	C	D	E	F	G	H
1								
2		(1)	(2)	(3)				
3		Price	Price	Price				
4								
5	Weight (lbs.)	1.747**	4.365***	5.716***				
6		(2.72)	(3.74)	(5.63)				
7								
8	Mileage (mpg)	-49.51	-86.79	-13.41				
9		(-0.57)	(-1.03)	(-0.19)				
10								
11	Length (in.)	-104.9*	-92.48**					
12		(-2.64)	(-2.75)					
13								
14	Car type		3550.2***					
15	-5.42							
16								
17	Constant	1946.1	14542.4*	5515.6				
18		(0.54)	(2.47)	(1.05)				
19								
20	Observations	74	74	74				
21	Adjusted R-squared	0.273	0.330	0.523				
22								
23	t statistics in parentheses							
24	* p<0.05, ** p<0.01, *** p<0.001							

2. estout 命令之 estout: 回归表格的较高质量输出

```
estout using "$PATH/estout_tab_2.xls", ///  
    stats( ///  
        r2_a N, ///  
        labels("Adj. R-sq." "Num of Obs") ///  
    ) label replace
```

	A	B	C	D
1		est1	est2	est3
2		b	b	b
3	Weight (lb	1.746559	4.364798	5.716181
4	Mileage (r	-49.5122	-86.7893	-13.4072
5	Length (in.)		-104.868	-92.4802
6	Car type			3550.194
7	_cons	1946.069	14542.43	5515.58
8	Adj. R-sq.	0.273485	0.329835	0.522952
9	Num of O	74	74	74
10				

- 这样输出的表格编辑和美化比较方便。
- 但是esttab中的星星丢失了。

2. estout 命令之 estout: 加入更多要素

```
estout using "$PATH/estout_tab_3.xls", ///
      cells(b(star)) ///
      stats(r2_a N, ///
            labels("Adj. R-sq." "Num of Obs")) ///
      postfoot("* p<0.05, ** p<0.01, *** p<0.001") ///
      label replace
```

	A	B	C	D	E
1		est1	est2	est3	
2		b	b	b	
3	Weight (lb)	1.746559*	4.364798*	5.716181***	
4	Mileage (r	-49.5122	-86.7893	-13.4072	
5	Length (in.)		-104.8682	-92.48018**	
6	Car type			3550.194***	
7	_cons	1946.069	14542.43*	5515.58	
8	Adj. R-sq.	0.273485	0.329835	0.522952	
9	Num of O	74	74	74	
10	* p<0.05, ** p<0.01, *** p<0.001				
11					

- 加入了星星和脚注。
- 更多选项参见: `help estout`

3. outreg2 命令

导出高质量的回归分析表格。

3. outreg2 命令

- 什么是outreg2? 官方介绍:
 - “outreg2提供了一种生成回归结果表格的快速简单的方法,所得到的表格以ASCII格式保存”。
 - “outreg2也报告一些常用统计值, 如: 以星号表示显著性水平的标准误、t统计量、p值、.....”
- outreg2也能输出描述分析表格。

- 安装:

```
ssc install logout, replace
```

```
ssc install outreg2, replace
```


3. outreg2 命令：回归表格的高质量输出

```
global PATH = "xxx"
sysuse auto, clear
eststo A: quietly regress price weight mpg
eststo B: quietly regress price weight mpg length
eststo C: quietly regress price weight mpg ///
           length foreign
outreg2 [A B C] using "$PATH/outreg2_tab.doc", ///
         replace tstat ///
         e(all) bdec(3) tdec(2) ctitle(price)
```

VARIABLES	(1) price	(2) price	(3) price
weight	1.747*** (2.72)	4.365*** (3.74)	5.716*** (5.63)
mpg	-49.512 (-0.57)	-86.789 (-1.03)	-13.407 (-0.19)
length		-104.868** (-2.64)	-92.480*** (-2.75)
foreign			3,550.194*** (5.42)
Constant	1,946.069 (0.54)	14,542.434** (2.47)	5,515.580 (1.05)
Observations	74	74	74
R-squared	0.293	0.357	0.549
df_m	2	3	4
df_r	71	70	69
F	14.74	12.98	21.01
r2	0.293	0.357	0.549
rmse	2514	2415	2037
mss	1.860e+08	2.270e+08	3.490e+08
rss	4.490e+08	4.080e+08	2.860e+08
r2_a	0.273	0.330	0.523
ll	-682.9	-679.4	-666.2
ll_0	-695.7	-695.7	-695.7
rank	3	4	5

t-statistics in parentheses
***p<0.01, **p<0.05, *p<0.1

3. outreg2 命令：回归表格的高质量输出

- [A B C]: 指定要用来制表的回归结果。
- replace: 覆盖同名文件。
- tstat: 将括号内默认的标准误改为t统计量。
- e(all): 报告所有可计算的返回值（样本量、R方、.....）。
 - 也可以用e(N r2_a)这样的写法指定报告哪些值（如图）。
- bdec(3) tdec(2): 指定回归系数和t统计量的小数位数。
- ctitle(price): 指定列标题。
- 还可以加入label选项，用标签代替变量名（如图）。
- 更多选项见: **help** outreg2

VARIABLES	(1)	(2)	(3)
	price	price	price
Weight (lbs.)	1.747*** (2.72)	4.365*** (3.74)	5.716*** (5.63)
Mileage (mpg)	-49.512 (-0.57)	-86.789 (-1.03)	-13.407 (-0.19)
Length (in.)		-104.868** (-2.64)	-92.480*** (-2.75)
Car type			3,550.194*** (5.42)
Constant	1,946.069 (0.54)	14,542.434** (2.47)	5,515.580 (1.05)
Observations	74	74	74
R-squared	0.293	0.357	0.549
N	74	74	74
r2_a	0.273	0.330	0.523
t-statistics in parentheses *** p<0.01, ** p<0.05, * p<0.1			

4. asdoc 命令

导出doc格式的各类表格。

4. asdoc 命令

- 什么是asdoc? 官方介绍:
 - “asdoc将Stata输出转换成Word/RTF格式”。
 - “使用asdoc非常简单, 我们只需要在Stata命令中添加asdoc作为前缀即可”。
- 安装: `ssc install asdoc, replace`
- 调用帮助文档: `help asdoc`

4. asdoc 命令：描述统计结果输出

```
global PATH = "xxx"
```

```
sysuse auto, clear
```

```
asdoc sum price mpg rep78 headroom, label replace save($PATH/asdoc_tab_1.doc)
```

- “label” 选项显示变量标签。

Descriptive Statistics					
Variable	Obs	Mean	Std. Dev.	Min	Max
Price	74	6165.257	2949.496	3291	15906
Mileage (mpg)	74	21.297	5.786	12	41
Repair Record 1978	69	3.406	.99	1	5
Headroom (in.)	74	2.993	.846	1.5	5
Trunk space (cu. ft.)	74	13.757	4.277	5	23

4. asdoc 命令：相关性分析结果输出

```
asdoc pwcorr price headroom mpg displacement, sig label replace ///  
      save($PATH/asdoc_tab_2.doc) title(Correlation Table)
```

- “sig” 选项显示显著性；“title()” 定制标题。

Correlation Table				
Variables	(1)	(2)	(3)	(4)
(1) Price	1.000			
(2) Headroom (in.)	0.115 (0.331)	1.000		
(3) Mileage (mpg)	-0.469 (0.000)	-0.414 (0.000)	1.000	
(4) Displacem.. in.)	0.495 (0.000)	0.474 (0.000)	-0.706 (0.000)	1.000

4. asdoc 命令：回归分析结果输出

```
asdoc reg price mpg rep78, save($PATH/asdoc_tab_3.doc) label nest replace
```

```
asdoc reg price mpg rep78 headroom, label nest append
```

```
asdoc reg price mpg rep78 headroom weight, label nest append
```

- “nest **append**” 选项：
 - 第一次用 “nest” ；
 - 之后用 “nest **append**” ；
 - 可以将多个回归结果聚合到一张表里。
- 更多功能见： **help** asdoc

...	.. (1)	.. (2)	.. (3)
... price	... price	... price	... price
· Mileage (mpg)	-271.643***	-289.346***	-57.676
..	(57.771)	(62.539)	(82.045)
· Repair Record 1978	666.957*	670.897*	857.039***
..	(342.356)	(343.521)	(314.77)
· Headroom (in.)		-300.029	-734.847*
..		(398.052)	(377.345)
· Weight (lbs.)			2.476***
..			(.634)
· _cons	9657.754***	10921.33***	-848.366
..	(1346.54)	(2153.003)	(3590.974)
· Observations	69	69	69
· R-squared	.251	.257	.4

Standard errors are in parentheses

***p<.01, **p<.05, *p<.1

5. sum2docx, corr2docx, reg2docx 命令

导出docx格式的描述统计/相关性分析/回归分析表格。

5. sum2docx, corr2docx, reg2docx 命令

- 由中国作者编写的第三方命令。使用非常简单。

- 安装:

```
ssc install sum2docx, replace
```

```
ssc install corr2docx, replace
```

```
ssc install reg2docx, replace
```

- 同样可以使用`help`命令查看帮助

Author

Chuntao LI
China Stata Club (爬虫俱乐部)
Wuhan, China
chtl@zuel.edu.cn

Yuan XUE
China Stata Club (爬虫俱乐部)
Wuhan, China
xueyuan19920310@163.com

5. sum2docx 命令

```
global PATH = "xxx"
sysuse auto, clear

sum2docx price-trunk ///
    using "$PATH/my_docx_1.docx", replace ///
    stats( ///
        N mean(%9.2f) sd min(%9.0g) ///
        median(%9.0g) max(%9.0g) ///
    ) title("表1: 描述统计")
```

5. sum2docx 命令

表1：描述统计

VarName	Obs	Mean	SD	Min	Median	Max
price	74	6165.26	2949.496	3291	5006.5	15906
mpg	74	21.30	5.786	12	20	41
rep78	69	3.41	0.990	1	3	5
headroom	74	2.99	0.846	1.5	3	5
trunk	74	13.76	4.277	5	14	23

5. sum2docx 命令

- `stats()`选项用于标注需要显示的统计量。
- 统计量后面的括号，用于标注字符串显示格式，如`mean(%9.2f)`。
 - `%9.2f`的含义为：显示宽度为9字符、保留小数后2位的浮点数；
 - `f`代表`float`，即浮点数；`g`则代表`general`，在数值较大时会采取科学计数法。
 - 更多关于字符串格式化的用法，参见`help format`。
- 更多选项参见`help sum2docx`。

5. corr2docx 命令

```
global PATH = "xxx"
sysuse auto, clear

corr2docx mpg weight length rep78 if foreign == 1 ///
    using "$PATH/my_docx_2.docx", replace ///
    star(* 0.05 ** 0.01) ///
    title("表2: 相关性分析") ///
    note("注: 仅统计进口汽车。* p<0.05 ** p<0.01")
```

5. corr2docx 命令

表2：相关性分析

	mpg	weight	length	rep78
mpg	1	-0.726**	-0.725**	0.058
weight	-0.621**	1	0.929**	0.382
length	-0.591**	0.912**	1	0.330
rep78	0.169	0.386	0.379	1

注：仅统计进口汽车。* $p<0.05$ ** $p<0.01$


5. corr2docx 命令

- `if foreign == 1: .*2docx`系列命令允许使用**if**语句对分析的范围进行限定。
- `star(* 0.05 ** 0.01)`: 标注星星和对应的显著性水平。
- `note(“注: 仅统计进口汽车。* $p < 0.05$ ** $p < 0.01$ ”)`: 添加脚注。
- 更多选项参见**help corr2docx**。

5. putdocx 命令

- 该命令可以新建docx文件，供导出使用。不需要专门安装。
- 下列代码可以新建文件并添加一个简单的标题。

```
global PATH = "xxx"
putdocx begin // 新建docx文件
putdocx paragraph // 添加段落
putdocx text ("表格汇总"), ///
    font("楷体", 14, black) bold linebreak // 写入文字
putdocx save "$PATH/表格汇总.docx", replace
```



表格汇总↓

5. putdocx 命令

- `putdocx text`有诸多选项，包括：
 - `nformat`: 指定数字格式;
 - `font`: 指定字体、字号和颜色;
 - `bold`和`italic`: 指定粗体和斜体;
 - `linebreak`: 在插入字符串后再加入一个换行符。
- 更多选项参见`help putdocx`。

5. putdocx 命令

- 将之前的代码中的replace改为append，文件名改为新的文件名，就可以将生成的表格都整合到新文档里。

```
sum2docx price-trunk ///
    using "$PATH/表格汇总.docx", append ///
    ...

corr2docx mpg weight length rep78 if foreign == 1 ///
    using "$PATH/表格汇总.docx", append ///
    ...
```


5. putdocx 命令

表格汇总↓

表1：描述统计

VarName	Obs	Mean	SD	Min	Median	Max
price	74	6165.26	2949.496	3291	5006.5	15906
mpg	74	21.30	5.786	12	20	41
rep78	69	3.41	0.990	1	3	5
headroom	74	2.99	0.846	1.5	3	5
trunk	74	13.76	4.277	5	14	23

表2：相关性分析

	mpg	weight	length	rep78
mpg	1	-0.726**	-0.725**	0.058
weight	-0.621**	1	0.929**	0.382
length	-0.591**	0.912**	1	0.330
rep78	0.169	0.386	0.379	1

注：仅统计进口汽车。* p<0.05 ** p<0.01

5. reg2docx 命令

```
global PATH = "xxx"
sysuse auto, clear

eststo A: quietly regress price weight mpg
eststo B: quietly regress price weight mpg length
eststo C: quietly regress price weight mpg length foreign
reg2docx A B C using "$PATH/表格汇总.docx", append ///
    b(%9.3f) t(%7.2f) scalars(N r2(%9.3f) r2_a(%9.2f)) ///
    mtitles("模型1" "模型2" "模型3") ///
    title("表3: 回归分析")
```

5. reg2docx 命令

- `b()` `t()`: 设置回归系数和t统计量的格式。
- `scalars()`: 设置要显示的统计量及其格式。
 - 类似于outreg2中的`e()`命令;
 - 类似于estout中的`stats()`命令。
- `mtitles()`: 设置列标题。
 - 参数的数量必须和输出的回归模型的数量一样。
- 更多选项参见`help reg2docx`。
- 该系列另有`t2docx`命令, 使用方式类似。

表格汇总↓



表1: 描述统计

VarName	Obs	Mean	SD	Min	Median	Max
price	74	6165.26	2949.496	3291	5006.5	15906
mpg	74	21.30	5.786	12	20	41
rep78	69	3.41	0.990	1	3	5
headroom	74	2.99	0.846	1.5	3	5
trunk	74	13.76	4.277	5	14	23

表2: 相关性分析

	mpg	weight	length	rep78
mpg	1	-0.726**	-0.725**	0.058
weight	-0.621**	1	0.929**	0.382
length	-0.591**	0.912**	1	0.330
rep78	0.169	0.386	0.379	1

注: 仅统计进口汽车。* p<0.05 ** p<0.01

表3: 回归分析

	(1)	(2)	(3)
	模型1	模型2	模型3
weight	1.747*** (2.72)	4.365*** (3.74)	5.716*** (5.63)
mpg	-49.512 (-0.57)	-86.789 (-1.03)	-13.407 (-0.19)
length		-104.868** (-2.64)	-92.480*** (-2.75)
foreign			3550.194*** (5.42)
_cons	1946.069 (0.54)	14542.434** (2.47)	5515.580 (1.05)
N	74	74	74
r2	0.293	0.357	0.549
r2 a	0.27	0.33	0.52


6. graph export 命令

导出图片。

6. graph export 命令

```
global PATH = "xxx"  
sysuse auto, clear  
twoway (scatter price length) (lfit price length), name(my_graph, replace)  
graph save my_graph "$PATH/my_graph.gph"  
graph export "$PATH/my_graph.png", name(my_graph) as(png) replace
```

- **graph save**命令输出gph格式文件，只有STATA能读取。
- **graph export**命令可以输出eps、png等多种格式图片文件。





点击图标进入GitHub，获取全部示例代码。